

FONCTIONS D'UNE VARIABLE RÉELLE : DÉRIVABILITÉ

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Les performances de Maple sont étonnantes dans le domaine de la dérivation. Ne serait-ce que pour cette raison, il faut utiliser ce logiciel. Ce chapitre présente d'abord les commandes basiques **diff** et **D**, puis la dérivation symbolique enfin traite la question de la dérivation des fonctions implicites.

Les deux modes de calcul des dérivées

On a vu dans le chapitre précédent qu'une fonction mathématique peut être définie soit à l'aide d'une expression, c'est à dire une formule indiquant comment est transformée une variable indépendante nominative, soit à l'aide d'une fonction-procédure spécifiant complètement le mode de correspondance entre un élément quelconque de l'ensemble de départ et un élément de l'ensemble d'arrivée. La dérivée fait donc logiquement appel à deux modes différents selon que la fonction mathématique est écrite sous forme d'une expression ou d'une fonction informatique.

Dérivées d'une expression d'une seule variable

Soit une expression **Xp**, dépendant explicitement de la variable **x**, dont on cherche les dérivées successives par rapport à cette variable. On commence par poser l'expression à dériver :

```
> restart;  
Xp:=ln(x);  
Xp := ln(x)
```

Dérivée première d'une expression

La dérivée première de l'expression **Xp** par rapport à **x** est donnée par la commande **diff(Xp,x)**. Le résultat peut être assigné mais le code d'assignation doit pas comporter d'apostrophe comme on le fait habituellement dans la notation newtonienne de la dérivée. Pour Maple, une apostrophe en appelle toujours une seconde; les deux encadrent une expression qui ne sera pas évaluée. Dans l'exemple suivant, Maple renvoie un message d'erreur et le curseur clignote à l'endroit où il y a une faute de syntaxe dans l'input.

```
> Xp' :=diff(Xp,x);  
Error, unexpected single forward quote
```

L'identificateur suivant est licite :

```
> Xpp:=diff(Xp,x);  
Xpp :=  $\frac{1}{x}$  (1.1.1.1)
```

On calculera la valeur prise par la dérivée en un point par simple évaluation, en utilisant **eval**, **subs** ou par assignation successive.

```
> eval(diff(Xp,x),x=1);eval(Xpp,x=2);#utilisation de eval  
subs(x=1,diff(Xp,x));subs(x=2,Xpp);#utilisation de subs  
x:=1;Xpp;x:=2;Xpp;#assignation successive de la
```


```

variable x
x:='x';#on "libère" la variable x
      1
      1/2
      1
      1/2
x:=1
      1
x:=2
      1/2
x:=x

```

On peut aussi demander une représentation graphique de la dérivée :

```
> plot(Xpp,x=1..5);
```



Si l'expression à dériver dépend elle-même d'une expression, alors Maple applique automatiquement les règles de dérivation d'une fonction composée.

```

> restart;
E:=ln(x);F:=arctan(E);F1:=diff(F,x);
      E := ln(x)
      F := arctan(ln(x))
      F1 := 1 / (x * (1 + ln(x)^2))

```

▼ Dérivées d'ordre supérieur d'une expression

Pour connaître les dérivées d'ordre supérieur d'une expression, on utilise la commande **diff(Xp,x,x,...,x)** dans laquelle **x** figure le nombre de fois nécessaire pour atteindre l'ordre de dérivation. Il existe une syntaxe abrégée équivalente qui fait appel au symbole de répétition séquentielle **\$**, à savoir **diff(Xp,x\$n)** où **n** est l'ordre de dérivation.

```

> restart;
P:=ln(x):
diff(P,x);diff(P,x,x);diff(P,x,x,x);#demande des

```

dérivées première, seconde et troisième de la fonction logarithme népérien


`diff(P,x$2);diff(P,x$3);diff(P,x$8);#même type de demande en utilisant le symbole $`

$$\frac{1}{x}$$
$$-\frac{1}{x^2}$$
$$\frac{2}{x^3}$$
$$-\frac{1}{x^2}$$
$$\frac{2}{x^3}$$
$$-\frac{5040}{x^8}$$

Bien entendu, on peut évaluer la valeur des dérivées en un point ou tracer les fonctions dérivées (à condition que la variable **x** soit libre) :

```
> restart:E:=ln(x):E1:=diff(E,x):E2:=diff(E,x$2):E5:=diff(E,x$5):  
x:=5:E1;E2;E5;#évaluation par assignation de x  
x:='x';#libération de x  
plot({E1,E2,E5},x=1..5);
```

$$\frac{1}{5}$$
$$-\frac{1}{25}$$
$$\frac{24}{3125}$$
$$x := x$$



▼ La commande inerte Diff

Afin de documenter les sorties Maple, on utilise des commandes dites **inertes** qui

suspendent l'évaluation d'un calcul. Elles s'utilisent avec la commande **value** pour amener un écho sous forme d'égalité où le membre de gauche indique l'opération de dérivation en écriture leibnizienne et le membre de droite le résultat du calcul de dérivation.

Le correspondant inerte de **diff** (écrit avec un d minuscule) est la commande **Diff** (écrit avec un d majuscule), qui renvoie simplement l'instruction donnée par l'utilisateur sous la forme d'un output Maple (écriture en 2D Output). Par exemple :

```
> Diff(ln(x/(x^2+1)),x);
```

$$\frac{d}{dx} \ln\left(\frac{x}{x^2+1}\right)$$

En forçant l'évaluation de **Diff**, on obtient le résultat qui aurait été donné directement par **diff** :

```
> Diff(ln(x/(x^2+1)),x)=value(Diff(ln(x/(x^2+1)),x));
```

$$\frac{d}{dx} \ln\left(\frac{x}{x^2+1}\right) = \frac{\left(\frac{1}{x^2+1} - \frac{2x^2}{(x^2+1)^2}\right)(x^2+1)}{x}$$

L'utilisation du dito abrège sensiblement l'écriture des commandes :

```
> Diff(ln(x/(x^2+1)),x):%=value(%);
```

$$\frac{d}{dx} \ln\left(\frac{x}{x^2+1}\right) = \frac{\left(\frac{1}{x^2+1} - \frac{2x^2}{(x^2+1)^2}\right)(x^2+1)}{x}$$

La syntaxe de **Diff** est la même que celle de **diff** :

```
> Diff(exp(-x^2),x$4):%=value(%);
```

$$\frac{d^4}{dx^4} e^{-x^2} = 12 e^{-x^2} - 48 x^2 e^{-x^2} + 16 x^4 e^{-x^2}$$

▼ Dérivées d'une fonction

L'objectif est ici de déterminer la fonction dérivée première d'une fonction donnée ainsi que ses fonctions dérivées successives.

On commence par définir la fonction.

```
> restart;
f:=x->(1+x^2)^(-1);
```

$$f := x \rightarrow \frac{1}{1+x^2}$$

▼ Dérivée première


On utilise la commande **D(f)** pour demander la dérivée première. On récupère alors en output une fonction qui permet d'évaluer immédiatement la valeur prise par la dérivée en un point ou de tracer sa courbe représentative :

```
> fp1:=D(f);
fp1(4);fp1(sqrt(2));#calculs de la dérivée première en
un point
plot(fp1,0..5);#représentation graphique
```

$$fp1 := x \rightarrow -\frac{2x}{(1+x^2)^2}$$

$$-\frac{8}{289}$$

$$-\frac{2}{9}\sqrt{2}$$



▼ Dérivée d'ordre supérieur

Pour déterminer les fonctions dérivées supérieures, utilise la commande **D(@@n)(f)**, où le symbole @@ exprime que l'opérateur de dérivation doit être utilisé plus d'une fois et **n** désigne l'ordre de dérivation. On récupère ainsi une fonction qu'on peut bien entendu utiliser pour évaluer une dérivée en un point ou tracer la courbe représentative.


```
> fp2:=(D@@2)(f);fp3:=(D@@3)(f);#fonctions dérivées
deuxième et troisième
fp2(2);fp3(2);#calculs de dérivées en un point
plot({fp2,fp3},0..1);#représentation graphique des
fonctions dérivées
```

$$fp2 := x \rightarrow \frac{8x^2}{(1+x^2)^3} - \frac{2}{(1+x^2)^2}$$

$$fp3 := x \rightarrow -\frac{48x^3}{(1+x^2)^4} + \frac{24x}{(1+x^2)^3}$$

$$\frac{22}{125}$$

$$-\frac{144}{625}$$



▼ Dérivation symbolique

Les économistes cherchent souvent les propriétés de fonctions qui ne sont pas explicitées, en particulier les caractéristiques de leurs dérivées successives (généralement, première et seconde). En tant que logiciel de calcul formel, Maple opère des dérivations symboliques sur des expressions ou des fonctions.

▼ Dérivation de fonctions générales

Soit f une fonction quelconque. Avec l'opérateur **diff**, Maple adopte la notation leibnitziennne pour exprimer les dérivées successives de l'expression correspondante :

```
> restart;diff(f(x),x);diff(F(x),x$2);diff(U(x),x$n);
```

$$\frac{d}{dx} f(x)$$

$$\frac{d^2}{dx^2} F(x)$$

$$\frac{d^n}{dx^n} U(x)$$

Les dérivées successives de la fonction proprement dite sont données avec les opérateurs **D** de dérivation et les outputs adoptent la notation mathématique rigoureuse :

```
> restart;D(f);(D@@2)(F);(D@@n)(U);
```

$$D(f)$$

$$D^{(2)}(F)$$

$$D^{(n)}(U)$$

Maple dérive sans problème les expressions dans lesquelles figurent un ou plusieurs paramètres. La seule obligation est de préciser en second argument de **diff** la variable par rapport à laquelle on dérive.

```
> diff(a*x+b,x); #fonction affine de x
a
```

```
> diff(a*(x+h),a); #fonction linéaire de a
```

$$x+h$$

```
> diff(a*x/(1+x^b),x);diff(a*x/(1+x^b),a);diff(a*x/(1+x^b),b);
```

$$\frac{a}{1+x^b} - \frac{a x^b b}{(1+x^b)^2}$$
$$\frac{x}{1+x^b}$$
$$- \frac{a x x^b \ln(x)}{(1+x^b)^2}$$

```
> diff(a^x,x); #exponentielle de base a
```

$$a^x \ln(a)$$

```
> diff(x^a,x); #fonction puissance
```

$$\frac{x^a a}{x}$$

Opérations sur les fonctions

Les propriétés usuelles de la dérivation symbolique sont appliquées par Maple, qu'on utilise **diff** ou **D**.

Commençons par l'addition de deux fonctions :

```
> restart;diff(f(x)+g(x),x);D(f+g);
```

$$\frac{d}{dx} f(x) + \frac{d}{dx} g(x)$$
$$D(f) + D(g)$$

Les instructions de dérivation **diff** et **D** appliquent le théorème : la dérivée d'une somme est la somme des dérivées.

Passons à la multiplication par un scalaire :

```
> restart;diff(lambda*f(x),x);D(lambda*f);assume(lambda,constant);D(lambda*f);
```

$$\lambda \left(\frac{d}{dx} f(x) \right)$$
$$D(\lambda) f + \lambda D(f)$$
$$\lambda \sim D(f)$$

Il y a une difficulté avec l'opérateur de dérivation **D** car Maple considère a priori que λ est une fonction au même titre que f et applique la règle de dérivation d'un produit de fonctions. Tout rentre dans l'ordre en signalant par **assume** que λ est une constante.

La dérivation d'un produit de fonctions ne pose aucun problème :

```
> restart;diff(f(x)*g(x),x);D(f*g);
```

$$\left(\frac{d}{dx} f(x) \right) g(x) + f(x) \left(\frac{d}{dx} g(x) \right)$$
$$D(f) g + f D(g)$$

tout comme la dérivation d'un quotient :

```
> restart;diff(f(x)/g(x),x);D(f/g);
```

$$\frac{\frac{d}{dx} f(x)}{g(x)} - \frac{f(x) \left(\frac{d}{dx} g(x) \right)}{g(x)^2}$$

$$\frac{D(f)}{g} - \frac{fD(g)}{g^2}$$

Enfin, pour la composition de deux fonctions, Maple utilise une syntaxe extrêmement rigoureuse :

```
> restart;diff(g(f(x)),x);D(g@f);
```

$$D(g)(f(x)) \left(\frac{d}{dx} f(x) \right)$$

$$D(g) @ f D(f)$$

Pour l'opérateur **diff**, l'output fait référence à ... **D**. Le résultat est le produit de la dérivée de la fonction g calculée au point $f(x)$, soit $D(g)(f(x))$ par la dérivée de f calculée au point x , soit $\frac{d}{dx} f(x)$.

Pour l'opérateur **D**, le résultat est le produit de la dérivée de la fonction g composée avec la fonction f , soit $D(g) @ f$, par la dérivée de f .

▼ Dérivée d'une fonction implicite

Considérons l'équation $f(x, y) = 0$ où la fonction f est de classe C^1 sur \mathbb{R}^2 . Si la dérivée partielle de f par rapport à x est différente de 0, alors, d'après le théorème des fonctions implicites, il existe une fonction Φ exprimant y en fonction de x dont la dérivée première est donnée par

$$\Phi'(x) = - \frac{\frac{\partial}{\partial x} f(x, y)}{\frac{\partial}{\partial y} f(x, y)}.$$

La commande **implicitdiff** donne la dérivée première et même les dérivées successives de Φ . Sa syntaxe est

implicitdiff(f(x,y),y(x),x) pour la détermination de la dérivée première de la fonction implicite Φ , et **implicitdiff(f(x,y),y(x),x\$n)** pour la dérivée d'ordre n de la fonction implicite Φ .

```
> implicitdiff(f(x,y),y(x),x);
```

$$-\frac{D_1(f)(x,y)}{D_2(f)(x,y)}$$

```
> convert(%,diff);#pour retrouver l'écriture usuelle en économie
```

$$-\frac{\frac{\partial}{\partial x} f(x, y)}{\frac{\partial}{\partial y} f(x, y)}$$

A titre d'exemple, considérons l'équation $x^2 + y^2 - 5xy + 4x + y = 0$. Cherchons les dérivées première, seconde et troisième de la fonction implicite donnant y en fonction de x .

```
> restart;
```

```
fonction:=x^2+y^2-5*x*y+4*x+y;
```

```
phil:=implicitdiff(fonction,y(x),x);
```



```
phi2:=implicitdiff(fonction,y(x),x,x);
```

```
phi3:=implicitdiff(fonction,y(x),x$3);
```

$$\text{fonction} := x^2 + y^2 - 5xy + 4x + y$$

$$\phi 1 := \frac{-2x + 5y - 4}{2y - 5x + 1}$$

$$\phi 2 := \frac{2(-37 + 21x^2 + 84x + 21y + 21y^2 - 105xy)}{-15x + 6y + 12y^2 + 8y^3 + 1 - 60y^2x + 150yx^2 - 60xy + 75x^2 - 125x^3}$$

$$\phi 3 := -\left(6(481 + 1491x^2 - 273y - 1869x + 1806xy - 273y^2 + 441x^3 + 441y^2x - 2205yx^2)\right) / \left(-5000y^2x^3 - 5000yx^3 - 200xy + 6250yx^4 - 400y^4x + 1500yx^2 + 2000y^3x^2 - 600y^2x + 3000y^2x^2 - 800y^3x + 10y - 25x + 1 - 1250x^3 + 250x^2 + 3125x^4 - 3125x^5 + 32y^5 + 80y^4 + 80y^3 + 40y^2\right)$$

La supériorité d'un logiciel tel que Maple sur la méthode "papier brouillon et crayon" est ainsi mise clairement en évidence dans les domaines de la fiabilité et de la vitesse de calcul.

Exercices

Exercice M.1 (dérivée logarithmique)

Soit f une fonction dont l'ensemble d'arrivée est inclus dans $]0; +\infty[$.

1. Donner sa dérivée logarithmique avec les opérateurs **diff** et **D**.
2. Soit $f(x) = x^a$. Donner sa dérivée logarithmique en $x = 2$. Quel opérateur vous semble le plus pratique?

Exercice E.1 (élasticités de demande et d'offre)

1. Soit f une fonction de la variable $x : y = f(x)$. Rappelez la définition de l'élasticité de y par rapport à x . Quelle est son interprétation économique?
2. La demande d'un produit dépend de son prix. Calculez l'élasticité-prix du bien quand la demande est une fonction affine puis quand elle est une fonction puissance. Dans chaque cas, représentez graphiquement la fonction de demande et l'élasticité prix correspondante.
3. L'offre d'un produit dépend de son prix. Calculez l'élasticité-prix du bien quand l'offre est une fonction affine puis quand elle est une fonction puissance. Dans chaque cas, représentez graphiquement la fonction d'offre et l'élasticité prix correspondante.

Exercice E.2 (taux marginal de substitution)

Soit la fonction d'utilité continûment différentiable $: (x, y) \rightarrow U(x, y)$, où x et y désignent les quantités consommées de deux biens. Le niveau d'utilité étant fixé, quel est le taux d'échange psychologique entre les deux biens?

Exercice E.3 (taux marginal de substitution technique)

Soit la fonction de production générale $: (l, k) \rightarrow F(l, k)$, où l et k désignent les quantités de travail et de capital utilisées. Le niveau de production étant fixé, Comment faut-il compenser la (petite) baisse d'un facteur pour maintenir constant le volume de production?